```
> restart:
> with(OreModules):
> with(OreMorphisms);
> with(linalg):
```

Let us consider another model of a tank containing a fluid and subjected to a one-dimensional horizontal move studied in N. Petit, P. Rouchon, "Dynamics and solutions to some control problems for water-tank systems", *IEEE Trans. Automatic Control*, 47 (2002), 595-609. The system matrix is defined by:

```
> A:=DefineOreAlgebra(diff=[d,t],dual_shift=[delta,s],polynom=[t,s],
> comm=[alpha]):
> R:=matrix(2,3,[d,-d*delta^2,alpha*d^2*delta,d*delta^2,-d,alpha*d^2*delta]);
```

$$R := \begin{bmatrix} d & -d\,\delta^2 & \alpha\,d^2\delta \\ d\,\delta^2 & -d & \alpha\,d^2\delta \end{bmatrix}$$

Let us consider the $A = \mathbb{Q}(\alpha)[d, \delta]$-module $M = A^{1\times 3}/(A^{1\times 2}\,R)$ finitely presented by the matrix R and let us compute the A-module structure of the endomorphism ring $E = \mathrm{end}_A(M)$ of M:

```
> Endo:=MorphismsConstCoeff(R,R,A):
```

The A-module E is finitely generated by the endomorphisms f_i's defined by $f_i(\pi(\lambda)) = \pi(\lambda\,P_i)$, where $\pi : A^{1\times 3} \longrightarrow M$ denotes the projection onto M, $\lambda \in A^{1\times 3}$ and P_i is one of the following matrices:

```
> Endo[1];
```

$$\left[\begin{bmatrix} 0 & 1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & -1 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ \delta^2 & -1 & \alpha\,d\,\delta \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 1-\delta^2 & 1-\delta^2 & 0 \end{bmatrix},\right.$$
$$\left.\begin{bmatrix} 0 & 0 & 0 \\ -1+\delta^2 & -1+\delta^2 & 0 \\ 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 \\ \alpha\,d & \alpha\,d & 0 \\ \delta & \delta & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & \alpha\,d\,\delta \\ 1 & -\delta^2 & 0 \\ 0 & 0 & -\delta^2-1 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 \\ 1 & -\delta^2 & \alpha\,d\,\delta \\ 0 & 0 & 0 \end{bmatrix}\right]$$

The generators f_i's of E satisfy the following A-linear relations

```
> Endo[2];
```

$$\begin{bmatrix} -d & 0 & d\,\delta^2 & 0 & 0 & 0 & d & 0 \\ d\,\delta^2 & 0 & -d & 0 & 0 & 0 & -d & 0 \\ 0 & d & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & d & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \delta & 0 & -1+\delta^2 & 0 & 0 \\ 0 & 0 & 0 & 0 & d & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & d \end{bmatrix}$$

i.e., if we denote by $F = (f_1 \;\ldots\; f_8)^T$, we then have $Endo[2]\,F = 0$.

The multiplication table $Endo[3]$ of the generators f_i's gives us a way to rewrite the composition $f_i \circ f_j$ in terms of A-linear combinations of the f_k's or, in other words, if we denote by $\otimes$ the Kronecker product, namely, $F \otimes F = ((f_1 \circ F)^T \;\ldots\; (f_8 \circ F)^T)^T$, then the multiplication table T of the generators f_j's satisfies $F \otimes F = T\,F$, where T is the matrix $Endo[3]$ without the first column which corresponds to the indices (i, j) of the product $f_i \circ f_j$. We do not print here this matrix as it belongs to $A^{64\times 8}$. We can use it for rewriting any polynomial in the f_i's with coefficients in A in terms of a A-linear combination of the generators f_j's.

Let us now try to compute idempotents of E defined by idempotent matrices, namely, elements $e \in E$ satisfying $e^2 = e$ and defined by matrices $P \in A^{3\times 3}$ and $Q \in A^{2\times 2}$ satisfying the relations $R\,P = Q\,R$, $P^2 = P$ and $Q^2 = Q$:

```
> Idem:=IdempotentsMatConstCoeff(R,Endo[1],A,0);
```

$$Idem := [[\begin{bmatrix} 1/2 & 1/2 & 0 \\ 1/2 & 1/2 & 0 \\ -c51\,(-1+\delta^2) & -c51\,(-1+\delta^2) & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix},$$

$$\begin{bmatrix} 0 & 0 & 0 \\ -\delta^2 & 1 & -\alpha\,\delta\,d \\ 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ \delta^2 & 0 & \alpha\,\delta\,d \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1/2 & -1/2 & 0 \\ -1/2 & 1/2 & 0 \\ -c51\,(-1+\delta^2) & -c51\,(-1+\delta^2) & 1 \end{bmatrix}],$$

$$[Ore\_algebra, [\text{“}diff\text{''}, dual\_shift], [t,s], [d,\delta], [t,s], [\alpha, c51], 0, [], [], [t,s], [], [], [diff = [d,t], dual\_shift = [\delta, s]]]]$$

Let us consider the first entry P_1 of $Idem[1]$ where we have set the arbitrary constant $c51$ to 0 and the matrix $Q_1 \in A^{2\times 2}$ satisfying $R\,P_1 = Q_1\,R$:

```
> P[1]:=subs(c51=0,evalm(Idem[1,1])); Q[1]:=Factorize(Mult(R,P[1],A),R,A);
```

$$P_1 := \begin{bmatrix} 1/2 & 1/2 & 0 \\ 1/2 & 1/2 & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad Q_1 := \begin{bmatrix} 1/2 & -1/2 \\ -1/2 & 1/2 \end{bmatrix}$$

As the entries of the matrices P_1 and Q_1 belong to $\mathbb{Q}$, using linear algebraic techniques, we can easily compute bases of the free A-modules $\ker_A(.P_1)$, $\ker_A(.Q_1)$, $\mathrm{im}_A(.P_1) = \ker_A(.(I_3 - P_1))$ and $\mathrm{im}_A(.Q_1) = \ker_A(.(I_2 - Q_1))$:

```
> U1:=SyzygyModule(P[1],A): U2:=SyzygyModule(evalm(1-P[1]),A):
> U:=stackmatrix(U1,U2);
> V1:=SyzygyModule(Q[1],A): V2:=SyzygyModule(evalm(1-Q[1]),A):
> V:=stackmatrix(V1,V2);
```

$$U := \begin{bmatrix} 1 & -1 & 0 \\ 0 & 0 & 1 \\ 1 & 1 & 0 \end{bmatrix} \quad V := \begin{bmatrix} 1 & 1 \\ 1 & -1 \end{bmatrix}$$

We can check that $J_1 = U\,P_1\,U^{-1}$ and $J_2 = V\,Q_1\,V^{-1}$ are block-diagonal matrices formed by the matrices 0_n and I_m:

```
> VERIF1:=Mult(U,P,LeftInverse(U,A),A);
> VERIF2:=Mult(V,Q,LeftInverse(V,A),A);
```

$$VERIF1 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad VERIF2 := \begin{bmatrix} 0 & 0 \\ 0 & 1 \end{bmatrix}$$

Then, the matrix R is equivalent to the following block-diagonal matrix $V\,R\,U^{-1}$:

```
> R_dec:=map(factor,simplify(Mult(V,R,LeftInverse(U,A),A)));
```

$$R\_dec := \begin{bmatrix} d\,(\delta^2+1) & 2\,\alpha\,d^2\delta & 0 \\ 0 & 0 & -d\,(\delta-1)\,(\delta+1) \end{bmatrix}$$

This last result can be directly obtained by means of the command *HeuristicDecomposition*:

```
> map(factor,HeuristicDecomposition(R,P[1],A)[1]);
```

$$\begin{bmatrix} d\left(\delta^2+1\right) & 2\,\alpha\,d^2\,\delta & 0 \\ 0 & 0 & -d\,(\delta-1)\,(\delta+1) \end{bmatrix}$$

We can use another idempotent matrix P_2 listed in $Idem[1]$ to obtain another decomposition of the matrix R. Let us consider the fourth one and the corresponding idempotent matrix Q_2:

```
> P[2]:=Idem[1,4]; Q[2]:=Factorize(Mult(R,P[2],A),R,A);
```

$$P_2 := \begin{bmatrix} 0 & 0 & 0 \\ -\delta^2 & 1 & -\alpha\,\delta\,d \\ 0 & 0 & 0 \end{bmatrix} \quad Q_2 := \begin{bmatrix} 0 & \delta^2 \\ 0 & 1 \end{bmatrix}$$

As we have $P_2^2 = P_2$ and $Q_2^2 = Q_2$, we know that the A-modules $\ker_A(.P_2)$, $\ker_A(.Q_2)$, $\mathrm{im}_A(.P_2) = \ker_A(.(I_3 - P_2))$ and $\mathrm{im}_A(.Q_2) = \ker_A(.(I_2 - Q_2))$ are projective, and thus, free by the Quillen-Suslin theorem. Let us compute basis of those free A-modules:

```
> U11:=SyzygyModule(P[2],A): U21:=SyzygyModule(evalm(1-P[2]),A):
> UU:=stackmatrix(U11,U21);
> V11:=SyzygyModule(Q[2],A): V21:=SyzygyModule(evalm(1-Q[2]),A):
> VV:=stackmatrix(V11,V21);
```

$$UU := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 0 & 1 \\ \delta^2 & -1 & \alpha\,\delta\,d \end{bmatrix} \quad VV := \begin{bmatrix} -1 & \delta^2 \\ 0 & 1 \end{bmatrix}$$

As previously, we can check that the idempotent matrices P_2 and Q_2 are equivalent to block-diagonal matrices formed by the matrices 0_n and I_m:

```
> VERIF1:=Mult(UU,P[1],LeftInverse(UU,A),A);
> VERIF2:=Mult(VV,Q[1],LeftInverse(VV,A),A);
```

$$VERIF1 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad VERIF2 := \begin{bmatrix} 0 & 0 \\ 0 & 1 \end{bmatrix}$$

Then, the matrix R is equivalent to the following block-diagonal matrix:

```
> R_dec1:=map(factor,simplify(Mult(VV,R,LeftInverse(UU,A),A)));
```

$$R\_dec1 := \begin{bmatrix} d\,(\delta-1)\,(\delta+1)\left(\delta^2+1\right) & \alpha\,d^2\,\delta\;(\delta-1)\,(\delta+1) & 0 \\ 0 & 0 & d \end{bmatrix}$$

We can check this last result by means of the command *HeuristicDecomposition*:

```
> map(factor,HeuristicDecomposition(R,P[2],A)[1]);
```

$$\begin{bmatrix} d\,(\delta-1)\,(\delta+1)\left(\delta^2+1\right) & \alpha\,d^2\,\delta\;(\delta-1)\,(\delta+1) & 0 \\ 0 & 0 & d \end{bmatrix}$$

Hence, we obtain another decomposition of the matrix R. If we denote by

$$\left\{\begin{array}{l} T_1 = (d\,(\delta^2+1) \quad 2\,\alpha\,d^2\,\delta), \\ T_2 = d\,(\delta^2-1), \\ T_3 = (d\,(\delta^2-1)\,(\delta^2+1) \quad \alpha\,d^2\,\delta\,(\delta^2-1)), \\ T_4 = d, \end{array}\right. \qquad \left\{\begin{array}{l} M_1 = A^{1\times 2}/(A\,T_1), \\ M_2 = A/(A\,T_2), \\ M_3 = A^{1\times 2}/(A\,T_3), \\ M_4 = A/(A\,T_4), \end{array}\right. \tag{1}$$

then we have the following decompositions of the A-module M:

$$M \cong M_1 \oplus M_2, \quad M \cong M_3 \oplus M_4. \tag{2}$$

Let us now study the A-module structure of E defined by $A^{1\times 8}/(A^{1\times 7}\, Endo[2])$:

```
> ext1:=Exti(Involution(Endo[2],A),A,1): ext1[1];
```

$$\begin{bmatrix} d\,\delta^2 - d & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & d & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & d\,\delta^2 - d & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & d & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & d & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & d\,\delta^2 - d & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & d \end{bmatrix}$$

```
> ext1[2];
```

$$\begin{bmatrix} 1 & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & \delta^2+1 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

Hence, the following torsion elements of E

$$\left\{ \begin{array}{l} t_1 = f_1 + f_3, \\ t_2 = f_2, \\ t_3 = (\delta^2+1)\, f_3 + f_7, \\ t_4 = f_4, \\ t_5 = f_5, \\ t_6 = f_6, \\ t_7 = f_8, \end{array} \right. \qquad \left\{ \begin{array}{l} d\,(\delta^2-1)\, t_1 = 0, \\ d\, t_2 = 0, \\ d\,(\delta^2-1)\, t_3 = 0, \\ d\, t_4 = 0, \\ d\, t_5 = 0, \\ d\,(\delta^2-1)\, t_6 = 0, \\ d\, t_7 = 0, \end{array} \right. \tag{3}$$

generate the A-module $t(E)$ and we have $E/t(E) = A^{1\times 8}/(A^{1\times 7}\, ext1[2])$. As the A-module $E/t(E)$ is torsion-free, it can be parametrized by means of the matrix $ext1[3]$ defined by

```
> ext1[3];
```

$$\begin{bmatrix} 1 \\ 0 \\ -1 \\ 0 \\ 0 \\ 0 \\ \delta^2+1 \\ 0 \end{bmatrix}$$

i.e., we have $E/t(E) \cong A^{1\times 8}\, ext1[3]$. As $ext1[3]$ admits a left-inverse over A defined by

```
> LeftInverse(ext1[3],A);
```

$$\begin{bmatrix} 0 & 0 & -1 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

we obtain that $A^{1\times 8}\, ext1[3] = A$, i.e., $E/t(E)$ is a free A-module of rank 1. Using that the short exact sequence of A-modules $0 \longrightarrow t(E) \xrightarrow{\iota} E \xrightarrow{\rho} E/t(E) \longrightarrow 0$ ends with a projective A-module, it splits and we get $E \cong t(E) \oplus E/t(E) \cong t(E) \oplus A$. Let us now study $t(E)$.

```
> L:=Factorize(Endo[2],ext1[2],A);
```

$$L := \begin{bmatrix} -d & 0 & d & 0 & 0 & 0 & 0 \\ d\,\delta^2 & 0 & -d & 0 & 0 & 0 & 0 \\ 0 & d & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & d & 0 & 0 & 0 \\ 0 & 0 & 0 & \delta & 0 & -1+\delta^2 & 0 \\ 0 & 0 & 0 & 0 & d & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & d \end{bmatrix}$$

```
> SyzygyModule(ext1[2],A);
```

$$INJ\,(7)$$

Lemma 3.1 of T. Cluzeau, A. Quadrat, "Factoring and decomposing a class of linear functional systems", *Linear Algebra and Its Applications*, 428 (2008), 324-381, we obtain that $t(E) \cong A^{1\times 7}/(A^{1\times 7}\, L)$. From the structure of the full row rank matrix L, we obtain that

$$t(E) \cong [A/(A\, d)]^3 \oplus A^{1\times 2}/(A^{1\times 2}\, S_1) \oplus A^{1\times 2}/(A^{1\times 2}\, S_2),$$

where where N^l denotes l direct sums of N and the matrices S_1 and S_2 are defined by:

```
> S[1]:=submatrix(L,1..2,[1,3]);
```

$$S_1 := \begin{bmatrix} -d & d \\ d\,\delta^2 & -d \end{bmatrix}$$

```
> S[2]:=submatrix(L,4..5,[4,6]);
```

$$S_2 := \begin{bmatrix} d & 0 \\ \delta & -1+\delta^2 \end{bmatrix}$$

Let us check whether or not the matrix S_1 is equivalent to a block-diagonal matrix:

```
> E[1]:=MorphismsConstCoeff(S[1],S[1],A):
> Idem[1]:=IdempotentsMatConstCoeff(S[1],E[1][1],A,0,alpha);
```

$$Idem_1 := [[\begin{bmatrix} c31 & -c31+1 \\ c31 & -c31+1 \end{bmatrix}, \begin{bmatrix} c31 & -c31 \\ c31-1 & -c31+1 \end{bmatrix}, \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}], [Ore\_algebra,$$
$$[\text{"}diff\text{"}, dual\_shift], [t,s], [d,\delta], [t,s], [\alpha, c31], 0, [], [], [t,s], [], [], [diff = [d,t], dual\_shift = [\delta, s]]]]$$

```
> X[1]:=subs(c31=0,evalm(Idem[1][1,1]));
```

$$X_1 := \begin{bmatrix} 0 & 1 \\ 0 & 1 \end{bmatrix}$$

We obtain that the matrix S_1 is equivalent to the following block diagonal matrix:

```
> map(factor,HeuristicDecomposition(S[1],X[1],A)[1]);
```

$$\begin{bmatrix} -d & 0 \\ 0 & d\,(\delta-1)\,(\delta+1) \end{bmatrix}$$

Hence, we have $A^{1\times 2}/(A^{1\times 2}\,S_1) \cong A/(A\,d) \oplus A/(A\,d\,(\delta^2-1))$.

Let us check whether or not the matrix S_2 is equivalent to a block-diagonal matrix:

```
> E[2]:=MorphismsConstCoeff(S[2],S[2],A):
```

```
> Idem[2]:=IdempotentsMatConstCoeff(S[2],E[2][1],A,0,alpha);
```

$$Idem_2 := [[\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}], [Ore\_algebra, [``diff", dual\_shift],$$
$$[t,s],[d,\delta],[t,s],[\alpha],0,[],[],[t,s],[],[],[diff = [d,t], dual\_shift = [\delta,s]]]]$$

```
> X[2]:=Idem[2][1,1];  Y[2]:=diag(0$2);  Z:=diag(0$2);
```

$$X_2 := \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix} \quad Y_2 := \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix} \quad Z := \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$$

```
> Lambda:=RiccatiConstCoeff(S[2],X[2],Y[2],Z,A,1,alpha)[1];
```

$$\Lambda := [[\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & \delta \\ 0 & -1 \end{bmatrix}]]$$

```
> X_bar[2]:=simplify(evalm(X[2]+Mult(Lambda[2],S[2],A)));
```

$$X\_bar_2 := \begin{bmatrix} \delta^2 & \left(-1+\delta^2\right)\delta \\ -\delta & 1-\delta^2 \end{bmatrix}$$

We obtain that the matrix S_2 is equivalent to the following block-diagonal one:

```
> map(factor,HeuristicDecomposition(S[2],X_bar[2],A)[1]);
```

$$\begin{bmatrix} d\,(\delta-1)\,(\delta+1) & 0 \\ 0 & 1 \end{bmatrix}$$

In particular, we have $A^{1\times 2}/(A^{1\times 2}\,S_2) \cong A/(A\,d\,(\delta^2-1))$, which shows that:

$$t(E) \cong [A/(A\,d)]^4 \oplus [A/(A\,d\,(\delta^2-1))]^2.$$

Hence, we obtain the following decomposition of the A-module E:

$$E \cong [A/(A\,d)]^4 \oplus [A/(A\,d\,(\delta^2-1))]^2 \oplus A. \tag{4}$$

We now explicitly describe the previous isomorphism. Let us first compute a generalized inverse of the matrix $ext1[2]$ over A:

```
> W:=GeneralizedInverse(ext1[2],A);
```

$$W := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

We now introduce the matrix $H = I_6 - W\, ext1[2]$ defined by:

```
> H:=simplify(evalm(1-Mult(W,ext1[2],A)));
```

$$H := \begin{bmatrix} 0 & 0 & -1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & -\delta^2 - 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

Using the fact that $ext1[2]\, H = 0$, we obtain that the A-morphism $\sigma : E/t(E) \longrightarrow E$ defined by $\sigma(\pi'(\lambda)) = \pi(\lambda\, H)$, where $\pi : A^{1\times 8} \longrightarrow E$ (resp., $\pi' : A^{1\times 8} \longrightarrow E/t(E)$) denotes the canonical projection onto E (resp., $E/t(E)$) and $\lambda \in A^{1\times 8}$, satisfies $\rho \circ \sigma = \mathrm{id}_{E/t(E)}$. For more details, see Theorem 4 of A. Quadrat, D. Robertz, "Parametrizing all solutions of uncontrollable multidimensional linear systems", *Proceedings of* 16^{th} *IFAC World Congress*, Prague (Czech Republic), 04-08/07/05. If we denote by $\{g_i = \rho(f_i)\}_{i=1,\ldots,8}$ a set of generators of the A-module $E/t(E)$, then the A-morphism $\sigma : E/t(E) \longrightarrow E$ is defined by:

$$\begin{cases} \sigma(g_1) = -f_3, \\ \sigma(g_2) = 0, \\ \sigma(g_3) = f_3, \\ \sigma(g_4) = 0, \\ \sigma(g_5) = 0, \\ \sigma(g_6) = 0, \\ \sigma(g_7) = -(\delta^2 + 1)\, f_3, \\ \sigma(g_8) = 0. \end{cases}$$

Using (3), the A-morphism $\chi : \mathrm{id}_E - \sigma \circ \rho : E \longrightarrow E$ is then defined by:

$$\begin{cases} \chi(f_1) = f_1 + f_3 = t_1, \\ \chi(f_2) = f_2 = t_2, \\ \chi(f_3) = f_3 - f_3 = 0, \\ \chi(f_4) = f_4 = t_4, \\ \chi(f_5) = f_5 = t_5, \\ \chi(f_6) = f_6 = t_6, \\ \chi(f_7) = f_7 + (\delta^2 + 1)\, f_3 = t_3, \\ \chi(f_8) = f_8 = t_7. \end{cases}$$

Hence, if we define the A-morphism $\kappa : E \longrightarrow t(E)$ by

$$\begin{cases} \kappa(f_1) = t_1, \\ \kappa(f_2) = t_2, \\ \kappa(f_3) = 0, \\ \kappa(f_4) = t_4, \\ \kappa(f_5) = t_5, \\ \kappa(f_6) = t_6, \\ \kappa(f_7) = t_3, \\ \kappa(f_8) = t_7, \end{cases}$$

then we get the identity $\mathrm{id}_E = \sigma \circ \rho + \iota \circ \kappa$. Therefore, we obtain:

$$\begin{cases} f_1 = t_1 - \mathrm{id}_M, \\ f_2 = t_2, \\ f_3 = \mathrm{id}_M, \\ f_4 = t_4, \\ f_5 = t_5, \\ f_6 = t_6, \\ f_7 = t_3 - (\delta^2 + 1)\, \mathrm{id}_M, \\ f_8 = t_7. \end{cases}$$

We find that $\{t_1, \ldots, t_7, \mathrm{id}_M\}$ is the same set of generators of the A-module E as $\{f_i\}_{i=1,\ldots,8}$. Hence, the family of generators $\{t_1, \ldots, t_7, \mathrm{id}_M\}$ admits the same multiplication table $Endo[3]$.

Let us show how to find again (4) from (2). Using the fact that $M \cong M_1 \oplus M_2$, we get:

$$E = \mathrm{end}_A(M) \cong \mathrm{end}_A(M_1) \oplus \mathrm{hom}_A(M_1, M_2) \oplus \mathrm{hom}_A(M_2, M_1) \oplus \mathrm{end}_A(M_2).$$

Using the fact that $M_2 = A/(A\, d\, (\delta^2 - 1))$, we have $\mathrm{end}_A(M_2) = A/(A\, d\, (\delta^2 - 1))$. With the notations (1)

```
> T[1]:=submatrix(R_dec,1..1,1..2);
```

$$T_1 := \left[\; d\left(\delta^2 + 1\right) \quad 2\,\alpha\, d^2 \delta \;\right]$$

```
> T[2]:=submatrix(R_dec,2..2,3..3);
```

$$T_2 := \left[\; -d\,(\delta - 1)\,(\delta + 1) \;\right]$$

we have $\mathrm{hom}_A(M_1, M_2) = A^{1\times 3}/(A^{1\times 3}\, Morph[1][2])$, where $Morph[1][2]$ is defined by:

```
> Morph[1]:=MorphismsConstMorphCoeff(T[1],T[2],A): Morph[1][2];
```

$$\begin{bmatrix} -1+\delta^2 & -\delta & 0 \\ 0 & d & 0 \\ 0 & 0 & d \end{bmatrix}$$

Using the structure of the matrix $Morph[1][2]$ and the previous decomposition of S_2, we obtain:

$$\hom_A(M_1, M_2) \cong A/(d\,(\delta^2-1)) \oplus A/(A\,d).$$

Let us now compute $\hom_A(M_2, M_1)$:

```
> Morph[2]:=MorphismsConstCoeff(T[2],T[1],A);
```

$$Morph_2 := [[\begin{bmatrix} \delta^2+1 & 2\,\alpha\,d\,\delta \end{bmatrix}], \begin{bmatrix} d \end{bmatrix}]$$

We obtain that $\hom_A(M_2, M_1)$ is generated by one generator h satisfying the relation $d\,h = 0$, i.e., we have $\hom_A(M_2, M_1) \cong A/(A\,d)$.

We now need to characterize the A-module $\mathrm{end}_A(M_1)$:

```
> Morph[3]:=MorphismsConstCoeff(T[1],T[1],A): Morph[3][2];
```

$$\begin{bmatrix} d & 0 & 0 \\ 0 & d\,\delta^2+d & d \end{bmatrix}$$

Hence, we obtain $\mathrm{end}_A(M_1) \cong A/(A\,d) \oplus A^{1\times 2}/(A\,J)$, where $J \in A^{1\times 2}$ is defined by:

```
> J:=submatrix(Morph[3][2],2..2,2..3);
```

$$J := \begin{bmatrix} d\,\delta^2+d & d \end{bmatrix}$$

Let us study the A-module $N = A^{1\times 2}/(A\,J)$:

```
> Extension1:= Exti(Involution(J,A),A,1);
```

$$Extension1 := [\begin{bmatrix} d \end{bmatrix}, \begin{bmatrix} \delta^2+1 & 1 \end{bmatrix}, \begin{bmatrix} -1 \\ \delta^2+1 \end{bmatrix}]$$

We get that $t(N) = (A\,((\delta^2+1) \quad 1))/(A\,J) \cong A/(A\,d)$ and $N/t(N) = A^{1\times 2}/(A\,((\delta^2+1) \quad 1))$. The A-module $N/t(N)$ is free as its parametrization $Extension1[3]$ admits a left-inverse over A:

```
> LeftInverse(Extension1[3],A);
```

$$\begin{bmatrix} -1 & 0 \end{bmatrix}$$

Therefore, the short exact sequence $0 \longrightarrow t(N) \longrightarrow N \longrightarrow N/t(N) \longrightarrow 0$ splits and we obtain that $N \cong t(N) \oplus N/t(N) \cong A/(A\,d) \oplus A$, a fact proving that $\mathrm{end}_A(M_1) \cong [A/(A\,d)]^2 \oplus A$ and:

$$\begin{aligned} E &\cong \mathrm{end}_A(M_1) \oplus \hom_A(M_1, M_2) \oplus \hom_A(M_2, M_1) \oplus \mathrm{end}_A(M_2) \\ &\cong [A/(A\,d)]^2 \oplus A \oplus A/(A\,d\,(\delta^2-1)) \oplus A/(A\,d) \oplus A/(A\,d) \oplus A/(A\,d\,(\delta^2-1)) \\ &\cong [A/(A\,d)]^4 \oplus [A/(A\,d\,(\delta^2-1))]^2 \oplus A. \end{aligned}$$

We can also use the second decomposition $M \cong M_3 \oplus M_4$ obtained in (2) to find again the previous result. Indeed, we have:

$$E = \mathrm{end}_A(M) \cong \mathrm{end}_A(M_3) \oplus \hom_A(M_3, M_4) \oplus \hom_A(M_4, M_3) \oplus \mathrm{end}_A(M_4).$$

Using similar techniques as the previous ones, we can prove that

$$\begin{cases} \mathrm{end}_A(M_3) \cong [A/(A\,d\,(\delta^2-1))]^2 \oplus A, \\ \hom_A(M_3, M_4) \cong [A/(A\,d)]^2, \\ \hom_A(M_4, M_3) \cong A/(A\,d), \\ \mathrm{end}_A(M_4) \cong A/(A\,d), \end{cases}$$

which finally shows again that $E \cong [A/(A\,d)]^4 \oplus [A/(A\,d\,(\delta^2-1))]^2 \oplus A$.